@窗口主题(兼容主题类型.兼容亮主题_无标题栏)
类 歌单窗口 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1]])
	变量 协调布局1 : 协调布局
	@布局配置([[父布局=协调布局1,宽度=-1,高度=-2,@显示组件到状态栏=真]])
	变量 应用栏布局1 : 应用栏布局
	@布局配置([[父布局=应用栏布局1,宽度=-1,@滚动=真,@折叠_最小高度=真,@吸附=真,高度DP=300]])
	变量 折叠工具栏1 : 折叠工具栏
	@布局配置([[父布局=折叠工具栏1,拉伸方式=6,高度=-1,宽度=-1,@视差=真,@显示组件到状态栏=真]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=折叠工具栏1,图片资源="../../素材/内置/png/bc4.xml",拉伸方式=6,高度=-1,宽度=-1,@视差=真,@显示组件到状态栏=真]])
	变量 图片框3 : 图片框
	@布局配置([[父布局=折叠工具栏1,宽度=-1,高度DP=50,@固定=真,标题=""]])
	变量 工具栏1 : 工具栏
	@布局配置([[父布局=工具栏1,宽度=-1,高度=-1,对齐方式=16]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,图片资源="../../素材/内置/png/返回.png",宽度DP=20,高度DP=20]])
	变量 图片框2 : 图片框
	@布局配置([[父布局=线性布局1,内容="音乐歌单标题",粗体=真,字体大小=17,字体颜色=-1,宽度DP=250,跑马灯效果=真,单行显示=真,可视=假]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-1,对齐方式=8388629]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,图片资源="../../素材/内置/png/q2.png",@右外边距DP=15,宽度DP=22,高度DP=22]])
	变量 图片框4 : 图片框
	@布局配置([[父布局=协调布局1,宽度=-1]])
	变量 高级列表框1 : 高级列表框

	@静态
	变量 歌单图片:文本
	@静态
	变量 歌单标题:文本
	@静态
	变量 歌单链接:文本

	变量 标题集合:文本集合
	变量 歌手集合:文本集合
	变量 图片集合:文本集合
	变量 链接集合:文本集合

	变量 网络工具1:网络工具
	事件 歌单窗口:创建完毕()
		订阅事件()
		本对象.导航栏颜色 = 文本到颜色值("#FFEEF1F6")
		协调布局1.启用透明状态栏()
		协调布局1.显示到状态栏()

		图片框1.加载网络图片(歌单图片)
		//文本框1.内容 = 歌单标题
		折叠工具栏1.标题 = 歌单标题
		折叠工具栏1.折叠后状态栏背景色 = "#FFEEF1F6"
		折叠工具栏1.折叠后背景颜色 = "#FFEEF1F6"

		网络工具1.取网页源码(公用数据.取歌单音乐列表接口+歌单链接)

		事件 网络工具1:取网页源码结束(结果 : 文本,内容 : 字节[],cookie : 文本)
			变量 歌单音乐结果对象:JSON对象=结果
			变量 歌单音乐项目数组:JSON数组=歌单音乐结果对象.取JSON对象("data").取JSON数组("musiclist")
			循环(音乐列表索引, 0, 歌单音乐项目数组.长度)
				图片集合.添加成员(歌单音乐项目数组.取JSON对象(音乐列表索引).取文本("img"))
				标题集合.添加成员(歌单音乐项目数组.取JSON对象(音乐列表索引).取文本("name"))
				歌手集合.添加成员(歌单音乐项目数组.取JSON对象(音乐列表索引).取文本("artist"))
				链接集合.添加成员(歌单音乐项目数组.取JSON对象(音乐列表索引).取文本("DC_TARGETID"))
			结束 循环
			变量 数据集 : 集合
			循环(i,,1)
				数据集.添加成员(i)
			结束 循环
			变量 适配器1 : 高级列表适配器 = 高级列表适配器(数据集)
			事件 适配器1:关联视图(项目类型 : 整数) : 组件容器
				变量 歌单音乐列表布局1:歌单音乐列表布局
				歌单音乐列表布局1.自定义列表框2.更新项目(1)
				事件 歌单音乐列表布局1.自定义列表框2:加载布局(索引 : 整数,项目布局 : 组件容器) : 组件容器
					变量 项目布局2:歌单列表顶部布局
					项目布局2.文本框1.内容 = 歌单标题
					项目布局2.文本框3.内容 = 标题集合.长度.到文本()
					项目布局2.文本框3.字体 = "fonts/2.ttf"
					返回 项目布局2
				结束 事件
				歌单音乐列表布局1.自定义列表框1.高度DP=(60*标题集合.长度)+85
				歌单音乐列表布局1.自定义列表框1.更新项目(标题集合.长度)
				事件 歌单音乐列表布局1.自定义列表框1:加载布局(索引 : 整数,项目布局 : 组件容器) : 组件容器
					变量 项目布局1:推荐页推荐布局新歌推荐1
					项目布局1.文本框3.内容 = (索引+1).到文本()
					项目布局1.文本框3.字体 = "fonts/1.ttf"
					假如 索引
						是 0
							项目布局1.图片框4.取图片().色调(文本到颜色值("#FE7C1F"))
						是 1
							项目布局1.图片框4.取图片().色调(文本到颜色值("#FFA51E"))
						是 2
							项目布局1.图片框4.取图片().色调(文本到颜色值("#FDCA00"))
						否则
							项目布局1.图片框4.取图片().色调(灰色)
					结束 假如
					项目布局1.文本框1.内容 = 歌手集合[索引]
					项目布局1.文本框2.内容 = 标题集合[索引]
					Glide图片加载.加载图片(项目布局1.图片框1,图片集合[索引])
					事件 项目布局1.图片框2:被单击()
						弹出提示("喜欢")
					结束 事件
					事件 项目布局1.图片框3:被单击()
						弹出提示("更多")
					结束 事件
					返回 项目布局1
				结束 事件
				事件 歌单音乐列表布局1.自定义列表框1:项目被单击(被单击组件 : 可视化组件,索引 : 整数)
					播放窗口.标题集合=标题集合
					播放窗口.图片集合=图片集合
					播放窗口.歌手集合=歌手集合
					播放窗口.音乐集合=链接集合
					播放窗口.音乐来源="酷我"
					播放窗口.当前播放索引=索引
					播放窗口.当前播放直连=假
					切换窗口(播放窗口)
					播放切换动画(自定义切换动画.从下向上_进入,自定义切换动画.从下向上_退出)
				结束 事件
				返回 歌单音乐列表布局1
			结束 事件
			高级列表框1.置适配器(适配器1)
			应用栏布局1.关联滚动组件(高级列表框1)
		结束 事件
		//0 为折叠状态, 1 为展开状态, 2 为中间状态
		事件 应用栏布局1:状态(状态 : 整数,距离 : 整数)
			假如 状态
				是 0
					本对象.状态栏字体黑色 = 真
					图片框2.取图片().色调(黑色)
					文本框1.字体颜色 = 黑色
					图片框4.取图片().色调(黑色)
					//文本框1.可视 = 真
				是 1
					本对象.状态栏字体黑色 = 假
					图片框2.取图片().色调(白色)
					文本框1.字体颜色 = 白色
					图片框4.取图片().色调(白色)
					//文本框1.可视 = 假
				是 2
					本对象.状态栏字体黑色 = 假
					图片框2.取图片().色调(白色)
					文本框1.字体颜色 = 白色
					图片框4.取图片().色调(白色)
					//文本框1.可视 = 假
			结束 假如
		结束 事件
	结束 事件

结束 类

类 歌单音乐列表布局 : 组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-2,背景颜色=-1117706]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-2,纵向布局=真,左内边距DP=18,右内边距DP=18]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,显示滑块条=假,始终显示快速滚动条=假,宽度=-1,高度DP=70]])
	变量 自定义列表框2 : 自定义列表框
	@布局配置([[父布局=线性布局1,显示滑块条=假,始终显示快速滚动条=假,宽度=-1]])
	变量 自定义列表框1 : 自定义列表框

结束 类

类 歌单列表顶部布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度DP=70,纵向布局=真,左内边距DP=0,右内边距DP=0,@下外边距DP=10]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,粗体=真,字体大小=17,字体颜色=-16777216,@下外边距DP=10]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局2,宽度=-1,高度DP=30]])
	变量 相对布局2 : 相对布局
	@布局配置([[父布局=相对布局2,高度=-1,对齐方式=16]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度DP=25,高度DP=25,@右外边距DP=5,图片资源="../../素材/内置/png/nav_scene_default.png"]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=线性布局3,@右外边距=8,内容="全部播放",粗体=真,字体颜色=-16777216,字体大小=14]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=线性布局3,字体大小=16,内容="0"]])
	变量 文本框3 : 文本框

结束 类